Вронскианная формула для КдФ

В.Э. Адлер. Классические интегрируемые системы, весенний семестр 2020

На лекции было получено семейство явных решений уравнения

$$u_t = u_{xxx} + 6\,u\,u_x$$

имеющее вид

$$u = 2\,\partial_x^2 \log W(y_1,\ \ldots,\ y_n),\quad y_j = c_j\,e^{-X_j} - e^{X_j},\ X_j = k_j\,x + 4\,k_j^3\,t.$$

Здесь мы проверим эту формулу для $n = 1, 2$ и посмотрим на графики при $n = 3$, при разных значениях параметров. Вычисление определителя – довольно затратная операция, поэтому при больших n формула малопригодна для расчетов (что не снижает ее теоретической значимости). Чуть более экономно вычисления проводятся при помощи преобразований Бэклунда, это мы еще будем проходить.

In[1]:=
```
W[f__] := Det[Table[D[{f}[[j]], {x, i - 1}], {i, 1, Length[{f}]}, {j, 1, Length[{f}]}]]
```

---

1 Вещественные k_j, c_j

In[2]:=
```
X[j_] := k[j] x + 4 k[j]^3 t
Y[j_] := c[j] Exp[-X[j]] - Exp[X[j]]
```

1.1 $n = 1$

In[4]:=
```
u1 = Simplify[2 D[Log[Y[1]], x, x]]
D[u1, t] - D[u1, x, x, x] - 6 u1 D[u1, x];
Simplify[%]
```

Out[4]= $-\dfrac{8\,e^{2\,x\,k[1]+8\,t\,k[1]^3}\,c[1]\,k[1]^2}{\left(e^{2\,x\,k[1]+8\,t\,k[1]^3}-c[1]\right)^2}$

Out[6]= 0

Построим график, считая k_1 и c_1 вещественными. При отрицательном c_1 имеем солитон, при положительном решение имеет полюс.

In[7]:=

```
Manipulate[Plot[u1 /. {k[1] → k1, c[1] → c1, t → tt}, {x, -10, 10},
  PlotRange → {-10, 10},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{k1, 1}, -2, 2, Appearance → "Labeled"},
 {{c1, -1}, -5, 5, Appearance → "Labeled"},
 {{tt, 0}, -5, 5, Appearance → "Labeled"}
]
```

Out[7]=



1.2 $n = 2$

Проверка формулы при $n = 2$ уже требует некоторого времени.

In[8]:=

```
u2 = Simplify[2 D[Log[W[Y[1], Y[2]]], x, x]]
D[u2, t] - D[u2, x, x, x] - 6 u2 D[u2, x];
Simplify[%]
```

Out[8]=

$$\Big(8\left(k[1]^2-k[2]^2\right)\Big(e^{2x(k[1]+2k[2])+8t(k[1]^3+2k[2]^3)}\,c[1]\,k[1]^2+e^{2xk[1]+8tk[1]^3}\,c[1]\,c[2]^2\,k[1]^2-e^{2x(2k[1]+k[2])+8t(2k[1]^3+k[2]^3)}\,c[2]\,k[2]^2-e^{2xk[2]+8tk[2]^3}\,c[1]^2\,c[2]\,k[2]^2-2\,e^{2x(k[1]+k[2])+8t(k[1]^3+k[2]^3)}\,c[1]\,c[2]\left(k[1]^2-k[2]^2\right)\Big)\Big)\Big/ \Big(e^{2x(k[1]+k[2])+8t(k[1]^3+k[2]^3)}\,(k[1]-k[2])+c[1]\,c[2]\,(-k[1]+k[2])+e^{2xk[2]+8tk[2]^3}\,c[1]\,(k[1]+k[2])-e^{2xk[1]+8tk[1]^3}\,c[2]\,(k[1]+k[2])\Big)^2$$

Out[10]= 0

При некоторых значениях параметров решение регулярно, при других имеет 1 или 2 полюса.

In[11]:=
```
Manipulate[Plot[u2 /. {k[1] → k1, k[2] → k2, c[1] → c1, c[2] → c2, t → tt}, {x, -10, 10},
  PlotRange → {-10, 10},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{k1, 1}, -2, 2, Appearance → "Labeled"},
 {{k2, 1.4}, -2, 2, Appearance → "Labeled"},
 {{c1, -1}, -5, 5, Appearance → "Labeled"},
 {{c2, 4}, -5, 5, Appearance → "Labeled"},
 {{tt, 0}, -5, 5, Appearance → "Labeled"}
]
```

Out[11]=



1.3 $n = 3$

Прямая проверка формулы при $n = 3$ безнадежна, так как само решение уже слишком громоздко (чтобы посмотреть на него, уберите точку с запятой после следующего присваивания и выполните его заново).

In[12]:=
```
u3 = Simplify[2 D[Log[W[Y[1], Y[2], Y[3]]], x, x]];
```

Тем не менее график все еще достаточно бодро строится. Опять, в зависимости от выбора параметров решение имеет от 0 до 3 полюсов.

In[13]:=
```
Manipulate[Plot[
   u3 /. {k[1] → k1, k[2] → k2, k[3] → k3, c[1] → c1, c[2] → c2, c[3] → c3, t → tt}, {x, -10, 10},
   PlotRange → {-10, 10},
   PlotStyle → {Blue},
   BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
   ImageSize → 300,
   AxesLabel → {"x", "u"}],
  {{k1, 1}, -2, 2, Appearance → "Labeled"},
  {{k2, 1.4}, -2, 2, Appearance → "Labeled"},
  {{k3, 1.9}, -2, 2, Appearance → "Labeled"},
  {{c1, -1}, -5, 5, Appearance → "Labeled"},
  {{c2, 4}, -5, 5, Appearance → "Labeled"},
  {{c3, -3}, -5, 5, Appearance → "Labeled"},
  {{tt, 0}, -5, 5, Appearance → "Labeled"}
]
```

Out[13]=



2 Пара примеров с комплексными параметрами

Вместо комбинации экспонент, можно брать в качестве y_j тригонометрические функции, что отвечает чисто мнимым k_j. Для примера, пусть y_1 экспоненты, а y_2 косинус.

In[14]:=
```
u4 = Simplify[2 D[Log[W[Y[1], Cos[k[2] x - 4 k[2]^3 t + d[2]]]], x, x]]
D[u4, t] - D[u4, x, x, x] - 6 u4 D[u4, x];
Simplify[TrigToExp[%]]
```

Out[14]=

$$\Big(2\left(k[1]^2+k[2]^2\right)\Big(2\,e^{2xk[1]+8tk[1]^3}c[1]\,\mathrm{Cos}\left[2\left(d[2]+x\,k[2]-4\,t\,k[2]^3\right)\right]k[1]^2-e^{4k[1]\left(x+4tk[1]^2\right)}k[2]^2-c[1]^2k[2]^2+2\,e^{2xk[1]+8tk[1]^3}c[1]\left(k[1]^2+k[2]^2\right)\Big)\Big)\Big/\Big(\left(e^{2xk[1]+8tk[1]^3}+c[1]\right)\mathrm{Cos}\left[d[2]+x\,k[2]-4\,t\,k[2]^3\right]k[1]+\left(e^{2xk[1]+8tk[1]^3}-c[1]\right)k[2]\,\mathrm{Sin}\left[d[2]+x\,k[2]-4\,t\,k[2]^3\right]\Big)^2$$

Out[16]= 0

In[17]:=
```
Manipulate[Plot[u4 /. {k[1] → k1, k[2] → k2, c[1] → c1, d[2] → d2, t → tt}, {x, -10, 10},
  PlotRange → {-10, 10},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{k1, 1}, -2, 2, Appearance → "Labeled"},
 {{k2, 0.5}, -2, 2, Appearance → "Labeled"},
 {{c1, 2}, -5, 5, Appearance → "Labeled"},
 {{d2, 0}, -5, 5, Appearance → "Labeled"},
 {{tt, 0}, -5, 5, Appearance → "Labeled"}
]
```

Out[17]=



Или возьмем пару комплексно сопряженных y_1, y_2. Получается такой волновой пакет с осцилляциями, но у него есть еще и полюс. В некоторых других уравнениях (например, в нелинейном уравнении Шредингера) возможны решения подобного типа без полюсов.

In[18]:=

```
u2 /. {c[1] → a + b ⅈ, c[2] → a - b ⅈ, k[1] → p + ⅈ q, k[2] → p - ⅈ q};
u5 = Simplify[ComplexExpand[%]]
```

Out[19]=

$$-\Big(\Big(16\, e^{2p\left(4p^2t-12q^2t+x\right)}\, p\, q\, \Big(4\left(a^2+b^2\right) e^{2p\left(4p^2t-12q^2t+x\right)}\, p\, q - \Big(2a^3pq + 2a\left(b^2+e^{4p\left(4p^2t-12q^2t+x\right)}\right) p q + b\left(-b^2+e^{4p\left(4p^2t-12q^2t+x\right)}\right)\left(p^2-q^2\right) + a^2 b\left(-p^2+q^2\right)\Big) \mathrm{Cos}\left[2q\left(12p^2t-4q^2t+x\right)\right] + \Big(2a^2bpq + 2b\left(b^2+e^{4p\left(4p^2t-12q^2t+x\right)}\right) p q + a^3\left(p^2-q^2\right) - a\left(-b^2+e^{4p\left(4p^2t-12q^2t+x\right)}\right)\left(p^2-q^2\right)\Big) \mathrm{Sin}\left[8\left(-3p^2q+q^3\right)t-2qx\right]\Big)\Big) \Big/ \Big(\Big(-a^2-b^2+e^{4p\left(4p^2t-12q^2t+x\right)}\Big) q + 2b\, e^{2p\left(4p^2t-12q^2t+x\right)}\, p\, \mathrm{Cos}\left[2q\left(12p^2t-4q^2t+x\right)\right] + 2a\, e^{2p\left(4p^2t-12q^2t+x\right)}\, p\, \mathrm{Sin}\left[8\left(-3p^2q+q^3\right)t-2qx\right]\Big)^2\Big)$$

In[20]:=

```
Manipulate[Plot[u5 /. {a → aa, b → bb, p → pp, q → qq, t → tt}, {x, -10, 10},
  PlotRange → {-10, 10},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "u"}],
 {{aa, 1}, -2, 2, Appearance → "Labeled"},
 {{bb, 0.5}, -2, 2, Appearance → "Labeled"},
 {{pp, 0.2}, -5, 5, Appearance → "Labeled"},
 {{qq, 3}, -5, 5, Appearance → "Labeled"},
 {{tt, 0}, -2, 2, Appearance → "Labeled"}
]
```

Out[20]=



3 Уравнение pot-КдФ

Уравнение pot-КдФ (потенцированное КдФ)

$$v_t = v_{xxx} + 6\, v_x^2$$

связано с КдФ подстановкой $u = 2\, v_x$ и для него вронскианная формула принимает чуть более простой вид

$$v = \partial_x \log W(y_1, \, \ldots, \, y_n), \quad y_j = c_j\, e^{-X_j} - e^{X_j}, \;\; X_j = k_j\, x + 4\, k_j^3\, t.$$

Легко понять, что солитонным решениям КдФ отвечают в этом уравнении решения, выходящие на разные постоянные при $x \to \pm\infty$. Такие решения называются кинками. Отметим, что есть разные типы кинков; например, в уравнении синус-Гордона кинки “топологические”, что означает, что предельные значения не могут быть произвольными, они “квантуются”. В уравнении pot-КдФ предельные значения могут быть любыми, но решение всегда возрастающее по x (не бывает антикинков).

Пример 2-кинкового решения.

In[21]:=
```
v2 = FullSimplify[D[Log[W[Sinh[X[1]], Cosh[X[2]]]], x]]
D[v2, t] - D[v2, x, x, x] - 6 D[v2, x]^2;
Simplify[%]
```

Out[21]=
$$\frac{k[1]^2 - k[2]^2}{\mathrm{Coth}\left[k[1]\,\left(x + 4\,t\,k[1]^2\right)\right]\,k[1] - k[2]\,\mathrm{Tanh}\left[k[2]\,\left(x + 4\,t\,k[2]^2\right)\right]}$$

Out[23]= 0

In[24]:=
```
Manipulate[
 Plot[v2 /. {k[1] → k1, k[2] → k2, d[1] → d1, d[2] → d2, t → tt}, {x, -60, 60},
  PlotRange → {-2.15, 2.15},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "v"}],
 {{k1, 1}, -2, 2, Appearance → "Labeled"},
 {{k2, 0.8}, -2, 2, Appearance → "Labeled"},
 {{d1, 0}, -2, 2, Appearance → "Labeled"},
 {{d2, 0.5}, -2, 2, Appearance → "Labeled"},
 {{tt, -12}, -12, 12, Appearance → "Labeled"}]
```

Out[24]=



3-кинковое решения.

In[25]:=
```
v3 = D[Log[W[Cosh[X[1]], Sinh[X[2]], Cosh[X[3]]]], x];
```

In[26]:=

```
Manipulate[
 Plot[
  v3 /. {k[1] → k1, k[2] → k2, k[3] → k3, d[1] → d1, d[2] → d2, d[3] → d3, t → tt}, {x, -60, 60},
  PlotRange → {-3.15, 3.15},
  PlotStyle → {Blue},
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 300,
  AxesLabel → {"x", "v"}],
 {{k1, 1}, -2, 2, Appearance → "Labeled"},
 {{k2, 0.8}, -2, 2, Appearance → "Labeled"},
 {{k3, 0.5}, -2, 2, Appearance → "Labeled"},
 {{d1, 0}, -2, 2, Appearance → "Labeled"},
 {{d2, 0.5}, -2, 2, Appearance → "Labeled"},
 {{d3, -0.5}, -2, 2, Appearance → "Labeled"},
 {{tt, -12}, -12, 12, Appearance → "Labeled"}]
```

Out[26]=